

BY THOMAS ROHR

## This month's Basic program saves time and effort in transistor amplifier calculations

Here's the first in a series of computer programs which will be simple to understand, easy to run on popular home computers and, usually, in beginner's Basic language. Our *Program of the Month* will show what a program looks like, how its steps are logical and, with study, how to write other programs for your personal computer. Our first package helps you calculate voltages, current, input resistances and the gain of transistor circuits.

The transistor amplifier circuit of figure 1 is perhaps the most common circuit configuration. It is an NPN transistor used in the common-emitter circuit to provide gain in audio and other low-frequency applications. Depending on the values of the resistors and capacitors used, it can provide gain in the range of five or less up to a hundred or more. It can operate at supply voltages from a few volts up to a hundred volts (with suitable transistors), at currents ranging from a fraction of a milliampere up to several tens of ma.

Our computer program, written in the Basic programming language used in home as well as professional computers, calculates the voltages and currents in the circuit, as well as the input resistance and the gain of the circuit. The voltages and currents are useful in troubleshooting, since they are easily measured with almost any inexpensive multimeter.

If you have a circuit to check, you can use this program to calculate the voltages and currents that it should have, and then compare these with the values actually measured. The gain, which describes how much larger the ac output voltage is than the input, can also be measured with fairly inexpensive equipment. The input resistance, while not especially useful for troubleshooting, is useful in deciding what can be connected to the amplifier's input.

Even if you do not have access to a computer, the program can still be interesting. By studying it, you can see that the Basic programming language is

### BASIC PROGRAM FOR TRANSISTOR AMPLIFIER CALCULATIONS

```

1 PRINT "ENTER VALUES OF VCC, R1, R2, R3, R4, R5 (IN OHMS)"
2 INPUT S, R1, R2, R3, R4, R5
3 PRINT "ENTER YOUR VALUE OF BETA"
4 INPUT B
5 V = S * R2 / (R1+R2)
6 R = (R1*R2) / (R1+R2)
7 I1 = (V-0.7) / (R + B*(R4+R5))
8 V1 = V - I1*R
9 I2 = I1*B
10 V2 = S - I2*R3
11 V3 = I2 * (R4+R5)
12 D = 0.025/I2 + 2
13 G = R3 / (R4+D)
14 T = B * (R4+D)
15 N = (R*T) / (R+T)
16 PRINT "BASE CURRENT IS "; I1*1000; " MILLIAMPERES"
17 PRINT "COLLECTOR CURRENT IS "; I2*1000; " MILLIAMPERES"
18 PRINT "EMITTER CURRENT IS "; I2*1000; " MILLIAMPERES"
19 PRINT "BASE VOLTAGE IS "; V1; " VOLTS"
20 PRINT "COLLECTOR VOLTAGE IS "; V2; " VOLTS"
21 PRINT "EMITTER VOLTAGE IS "; V3; " VOLTS"
22 PRINT "INPUT RESISTANCE IS "; N; "OHMS"
23 PRINT "THE GAIN OF THE AMPLIFIER IS "; G
24 PRINT
25 GO TO 3
26 END
    
```




Figure 1: schematic diagram of amplifier

quite easy and straightforward to use and understand.

As most programs, this one has a short section at the beginning which asks for the user to enter the values of the supply voltage, the five resistors, and the beta of the transistor (also called  $h_{fe}$  in the transistor spec sheets.) The center section of the program calculates the results, while the last section of the program prints them out in an orderly format.

As it turns out, the circuit's operation does not depend very much on the beta of the transistor at all. This is one of the reasons why it is so popular. To illustrate this point, the program has an instruction at the end which says *GO TO 3*. This returns the computer back to line 3, and allows the same set of calculations to be repeated, but with a different value of beta. A sample set of calculations is shown beneath the program listing.

If you want to use the formulas without having a computer, the following calculations are done. Start with these given values:

- S = value of dc supply voltage (also called  $V_{cc}$ )
- $R_1$  = top base bias resistor (see figure 1)
- $R_2$  = bottom base bias resistor
- $R_3$  = collector resistor
- $R_4$  = emitter resistor which is unby-passed by capacitor
- $R_5$  = emitter resistor which is bypassed
- B = transistor beta.

If the Beta is not known, a good trial value is 100 for small transistors. If no emitter resistor is present at all, use 0 ohms for  $R_4$  and  $R_5$ . If the entire emitter resistance is bypassed by a capacitor use 0 ohms for  $R_4$ . If no emitter bypass capacitor is used, then use 0 ohms for  $R_5$ .

Each line of the Basic program has a line number, which identifies the line to the computer. Lines 5, 6, and 7 are performed, in that order, to find  $I_b$ , the base current. Arithmetic formulas in Basic are written in the same way as they might be written in algebra, except that they must be written all on one line. Thus line 5 could also be written as

$$V = \frac{S R_2}{R_1 + R_2}$$

but the one-line form used in the program is equally easy to understand. The only strange symbol is \*, which means 'times'.

To solve the problem without a computer, the equations are done in order as listed in the program. After the base current is found in line 7, base voltage  $V_1$  is found in line 8, followed by collector current in line 9.

The collector and emitter voltages are found next in lines 10 and 11. Following this, lines 12 and 13 are done to find the gain G, followed by lines 14 and 15 to find the input resistance N.

The calculated values then are printed

by the computer using lines 16 through 23. In each case, a *PRINT* instruction tells the computer to start a new line, print the words enclosed in parentheses exactly as shown, followed by the value of the calculated result. Since the currents were found in amperes, each current value is multiplied by 1000 in the *PRINT* instruction to convert it to milliamperes.

After the last printout in line 23, line 24 says *PRINT* one more time, but without specifying anything to be printed. This is used to skip a line in the printout. Then the program says *GO TO 3*,

which instructs the computer to return to line 3 and start again from there. This allows you to keep the same resistor and voltage values specified at the beginning, but enter a different value of beta. The computer will repeat the program as long as you wish, allowing you to try different values of beta each time.

The printout of the results shows the values calculated by the program for the typical circuit of Figure 2. As you can see, doubling the value of beta, from 50 to 100, has very little effect on the results. Only the input resistance changes markedly.




Figure 2: sample circuit

## SAMPLE RESULTS FOR CIRCUIT OF FIGURE 2

```
ENTER VALUES OF VCC, R1, R2, R3, R4, R5 (IN OHMS)
?20, 27000, 10000, 1000, 47, 470
ENTER YOUR VALUE OF BETA
?50
BASE CURRENT IS .141954 MILLIAMPERES
COLLECTOR CURRENT IS 7.09772 MILLIAMPERES
EMITTER CURRENT IS 7.09772 MILLIAMPERES
BASE VOLTAGE IS 4.36952 VOLTS
COLLECTOR VOLTAGE IS 12.9023 VOLTS
EMITTER VOLTAGE IS 3.66952 VOLTS
INPUT RESISTANCE IS 1931.14 OHMS
THE GAIN OF THE AMPLIFIER IS 19.0396
```

```
ENTER YOUR VALUE OF BETA
?100
BASE CURRENT IS 7.97563E-02 MILLIAMPERES
COLLECTOR CURRENT IS 7.97563 MILLIAMPERES
EMITTER CURRENT IS 7.97563 MILLIAMPERES
BASE VOLTAGE IS 4.8234 VOLTS
COLLECTOR VOLTAGE IS 12.0244 VOLTS
EMITTER VOLTAGE IS 4.1234 VOLTS
INPUT RESISTANCE IS 3040.91 OHMS
THE GAIN OF THE AMPLIFIER IS 19.1811
```